#/*****************************************************************************
#* Keppe Messenger (KMP) Keppe Project-Team User Messaging System
#* Copyright (C) 2013-2014 Keppe Project-Team
#*
#* This program is free software: you can redistribute it and/or modify
#* it under the terms of the GNU General Public License as published by
#* the Free Software Foundation, either version 3 of the License, or
#* (at your option) any later version.
#*
#* This program is distributed in the hope that it will be useful,
#* but WITHOUT ANY WARRANTY; without even the implied warranty of
#* MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
#* GNU General Public License for more details.
#*
#* You should have received a copy of the GNU General Public License
#* along with this program.  If not, see <http://www.gnu.org/licenses/>.
#*****************************************************************************/

QT       += core gui network
isEqual(QT_MAJOR_VERSION, 5): QT += widgets multimedia

isEqual(QT_MAJOR_VERSION, 4): DEFINES += "QT4=0x001"
isEqual(QT_MAJOR_VERSION, 5): DEFINES += "QT5=0x001"
#win32: DEFINES += "ALPHA_UPDATE_MODE=0x001"
*msvc*: DEFINES += "MSVC=0x001"
*g++*: DEFINES += "GCC=0x001"
win64: DEFINES += "WIN64=0x001"
unix: DEFINES += "UNIX=0x001"
macx: DEFINES += "MACX=0x001"

shared{
DEFINES += "SHARED=0x001"
win32: RC_FILE = app.rc
win32: DEFINES += WIN32_SHARED
win32: DEFINES += WIN32MAC_SHARED
macx: DEFINES += MAC_SHARED
macx: DEFINES += WIN32MAC_SHARED
}
static{
DEFINES += "STATIC=0x001"
win32: RC_FILE = app.rc
win32: DEFINES += WIN32_STATIC
}

DEPLOYMENT.display_name = Keppe Messenger

VERSION = 1.0.0.30

win32: TARGET = KeppeMessenger
unix: TARGET = keppemessenger
macx: ICON = km.icns
macx: TARGET = "Keppe Messenger"

TEMPLATE = app

SOURCES += main.cpp\
    frmAboutKeppe.cpp \
    frmAboutKMPQ.cpp \
    frmChangePassword.cpp \
    frmChat.cpp \
    frmConfig.cpp \
    frmContactRequests.cpp \
    frmLogin.cpp \
    frmProfile.cpp \
    frmProfileInfo.cpp \
    NotificationWidget.cpp

HEADERS  += frmAboutKeppe.h \
    frmAboutKMPQ.h \
    frmChangePassword.h \
    frmChat.h \
    frmConfig.h \
    frmContactRequests.h \
    frmLogin.h \
    frmProfile.h \
    frmProfileInfo.h \
    NotificationWidget.h

FORMS += frmAboutKeppe.ui \
    frmAboutKMPQ.ui \
    frmChangePassword.ui \
    frmChat.ui \
    frmConfig.ui \
    frmContactRequests.ui \
    frmLogin.ui \
    frmProfile.ui \
    frmProfileInfo.ui \
    NotificationWidget.ui

shared{
win32: SOURCES += frmServerChat.cpp \
    frmUpdate.cpp \
    UpdateThread.cpp
win32: HEADERS += frmServerChat.h \
    frmUpdate.h \
    UpdateThread.h
win32: FORMS += frmServerChat.ui \
    frmUpdate.ui
macx: SOURCES += frmUpdate.cpp \
    UpdateThread.cpp
macx: HEADERS += frmUpdate.h \
    UpdateThread.h
macx: FORMS += frmUpdate.ui
}

OTHER_FILES = app.rc \
    res/qt_de.ts \
    res/credits.html

RESOURCES += \
    res/res.qrc

TRANSLATIONS += \
    res/qt_de.ts

INCLUDEPATH += ../includes
win32: LIBS += -L$$OUT_PWD/../KMsgFw/release -L$$OUT_PWD/../KGuiFw/release -llibkmsgfw1 -llibkguifw1
shared: macx: LIBS += -F$$OUT_PWD/../KMsgFw -F$$OUT_PWD/../KGuiFw -framework kmsgfw -framework kguifw
static: macx: LIBS += -L$$OUT_PWD/../KMsgFw -L$$OUT_PWD/../KGuiFw -lkmsgfw -lkguifw
unix: !macx: LIBS += -L$$OUT_PWD/../KMsgFw -L$$OUT_PWD/../KGuiFw -lkmsgfw -lkguifw

unix: !macx: scripts.path = /usr/local/bin
unix: !macx: scripts.files = $$PWD/bin/*
unix: !macx: target.path = /opt/keppemessenger/bin
unix: !macx: INSTALLS += target scripts
